

**Use ODS Excel, ODS PDF, ODS HTML5, ODS LAYOUT**

**LeRoy Bessler PhD**

**Bessler Consulting and Research**

**BesslerLeRoy2024@gmail.com**

**Visual Data Insights™**

**Strong Smart Systems™**

**Valparaiso, Indiana, USA**

LeRoy Bessler PhD is a data artist, the world's longest serving and most frequent advisor to SAS<sup>®</sup> users on best practices for graphic design and use of color, and the author of *Visual Data Insights Using SAS<sup>®</sup> ODS Graphics: A Guide to Communication-Effective Data Visualization*.

He also has specialized in Software-Intelligent Application Development for reliability, reusability, maintainability, flexibility, and extendibility, as well as ODS tools (Excel, PDF, HTML5, LAYOUT) to deliver SAS output and reports, tabular and/graphic. LeRoy is a data analyst and SAS programmer, who has supported SAS servers, SAS software, SAS analytic data, and hundreds of SAS users at employers and client sites.

He is a quarterly contributor to the *VIEWS News* online newsletter. He has presented at SAS users conferences in the USA, Canada, Ireland, England, and Germany, and has provided pre- and post-conference training. An advocate for, and provider of, SAS user mutual education opportunities, LeRoy was event organizer for the Wisconsin Illinois SAS Users Group for most years from 1989 to 2016. For the MidWest SAS Users Group he served as Conference Chair in 1991, Assistant to Co-chairs in 1993, and Program Chair in 2010.

# Presentation Format & Content

- ▶ This is a collection of excerpts from my WUSS 2024 training course, with updates and enhancements.
- ▶ It's structured as three presentations.
- ▶ ODS LAYOUT is presented by example, but not as a separate topic.

# **Get Your Report to the ODS Excel Destination**

LeRoy Bessler PhD

Bessler Consulting and Research

BesslerLeRoy2024@gmail.com

Visual Data Insights™

Strong Smart Systems™

Valparaiso, Indiana, USA

# Presentation Content

Spreadsheets for the screen captures are from Microsoft Excel 2010

**NOTE:** The examples are created at various levels of Zoom. The percent Zoom selected was chosen to fill the screen on my auxiliary monitor before the screen capture. If you re-create the output, it might display differently than what you see in these slides. Remove the ZOOM option, or adjust it as you prefer.

Any graphs used here are for proof of concept, and are not necessarily meant as good graphic design

# Common Frame for All ODS Excel Examples Used Here

```
title;footnote; /* "erase any from prior run in this session */
ods results off; /* not opening result in this session */
ods _all_ close;
options nocenter; /* override the default */
ods excel file="C:\MWSUG 2025\ODS Output\ExampleN.xlsx"
      options (name1='value1'
              name2='value2'...);
< SAS code goes here >
ods excel close;
ods results on; /* so later processing can open other results in
this same SAS session, unless results is again turned off */
```

# Simple Example

```
ods excel file="C:\MWSUG 2025\ODS Output\Example1.xlsx"
  options ( embedded_titles='yes' zoom='200' );
title1 justify=left /* JUSTIFY=CENTER is the default */
color=black
'Example 1 - Shoe Sales - First 10 Observations (200% Zoom) '
color=red ' See the default sheet name (maximum 28 characters) '
color=blue ' This title has wrapped.';
/* You can change colors within the title string */
options obs=10;
proc print data=sashelp.shoes noobs;
id region subsidiary product;
run;
ods excel close;
```

# Default Sheet Name & TITLE1 Wrap

The screenshot shows the Microsoft Excel interface. The title bar reads "Example1.xlsx - Microsoft Excel". The ribbon includes File, Home, Insert, Page Layout, Formulas, Data, Review, and View. The Home ribbon is active, showing options for Font, Alignment, Number, Styles, Cells, and Editing. The formula bar shows "A1" and the text "Example 1 - Shoe Sales - First 10 Observations (200% Zoom) See the default sheet name (maximum 28 characters) This title has wrapped." The worksheet grid shows a title in cell A1 that is wrapped across columns A through G. Below the title is a table with 7 columns: Region, Subsidiary, Product, Stores, Sales, Inventory, and Returns. The data rows are numbered 3 through 12.

	A	B	C	D	E	F	G	H
1	<b>Example 1 - Shoe Sales - First 10 Observations (200% Zoom) See the default sheet name (maximum 28 characters) This title has wrapped.</b>							
2								
3	<b>Region</b>	<b>Subsidiary</b>	<b>Product</b>	<b>Stores</b>	<b>Sales</b>	<b>Inventory</b>	<b>Returns</b>	
4	Africa	Addis Ababa	Boot	12	\$29,761	\$191,821	\$769	
5	Africa	Addis Ababa	Men's Casual	4	\$67,242	\$118,036	\$2,284	
6	Africa	Addis Ababa	Men's Dress	7	\$76,793	\$136,273	\$2,433	
7	Africa	Addis Ababa	Sandal	10	\$62,819	\$204,284	\$1,861	
8	Africa	Addis Ababa	Slipper	14	\$68,641	\$279,795	\$1,771	
9	Africa	Addis Ababa	Sport Shoe	4	\$1,690	\$16,634	\$79	
10	Africa	Addis Ababa	Women's Casual	2	\$51,541	\$98,641	\$940	
11	Africa	Addis Ababa	Women's Dress	12	\$108,942	\$311,017	\$3,233	
12	Africa	Algiers	Boot	21	\$21,297	\$73,737	\$710	

# Custom sheet\_name and NO title wrap

```
ods excel file="C:\MWSUG 2025\ODS Output\Example2.xlsx"
  options ( embedded_titles='yes' zoom='200'
           sheet_name='Custom Sheet Name' /* customize it */
           title_footnote_nobreak='yes' ); /* stop the wrap */
title1 justify=left
color=black 'Example 2 - Custom Sheet Name' color=blue
' NON-wrapped title extends beyond columns for data.';
options obs=10;
proc print data=sashelp.shoes noobs;
id region subsidiary product;
run;
ods excel close;
```



# TITLE1 as a hyperlink to elsewhere

```
ods excel nogtitle /* enable hyperlinks */
  file="C:\MWSUG 2025\ODS Output\Example3.xlsx"
  options ( embedded_titles='yes' zoom='200'
           sheet_name='Includes a HyperLink'
           title_footnote_nobreak='yes');
title1 justify=left color=blue underlin=1
/* COLOR and UNDERLIN are optional signals that it is a link */
  link='https://www.MWSUG.org'
  'Example 3. Click here to go to the MWSUG web home page';
options obs=10;
proc print data=sashelp.shoes noobs;
id region subsidiary product; run;
ods excel close;
options obs=max; /* for next run during this session */
```

# TITLE1 As a HyperLink to Elsewhere

ExcelExample3.xlsx - Microsoft Excel

File Home Insert Page Layout Formulas Data Review View

Clipboard Font Alignment Number Styles Cells Editing

A15

**[Example 3. Click here to go to the MWSUG web home page](#)**

Region	Subsidiary	Product	Stores	Sales	Inventory	Returns
Africa	Addis Ababa	Boot	12	\$29,761	\$191,821	\$769
Africa	Addis Ababa	Men's Casual	4	\$67,242	\$118,036	\$2,284
Africa	Addis Ababa	Men's Dress	7	\$76,793	\$136,273	\$2,433
Africa	Addis Ababa	Sandal	10	\$62,819	\$204,284	\$1,861
Africa	Addis Ababa	Slipper	14	\$68,641	\$279,795	\$1,771
Africa	Addis Ababa	Sport Shoe	4	\$1,690	\$16,634	\$79
Africa	Addis Ababa	Women's Casual	2	\$51,541	\$98,641	\$940
Africa	Addis Ababa	Women's Dress	12	\$108,942	\$311,017	\$3,233
Africa	Algiers	Boot	21	\$21,297	\$73,737	\$710
Africa	Algiers	Men's Casual	4	\$63,206	\$100,982	\$2,221

Includes a HyperLink

Ready

# Freeze Column Headers & Row Headers

```
ods excel file="C:\MWSUG 2025\ODS Output\Example4.xlsx"
  options ( embedded_titles='yes' zoom='200'
    title_footnote_nobreak='yes'
    frozen_headers='3' frozen_rowheaders='3'
    sheet_name='Freeze Headers + RowHeaders' );
  /* & in sheet_name would get converted to - */
title1 justify=left color=black 'Example 4' color=blue
' Freezing 3 rowheaders and ALL columnheaders through Row 3';
options obs=10;
proc print data=sashelp.shoes noobs;
id region subsidiary product; run;
ods excel close;
options obs=max; /* for next run during this session */
```

# Frozen Column Headers & Row Headers

ExcelExample4.xlsx - Microsoft Excel

File Home Insert Page Layout Formulas Data Review View

Clipboard Font Alignment Number Styles Cells Editing

D4 12

	A	B	C	D	E	F	G	H
1	<b>Example 4 Freezing 3 rowheaders and ALL columnheaders through Row 3</b>							
2								
3	Region	Subsidiary	Product	Stores	Sales	Inventory	Returns	
4	Africa	Addis Ababa	Boot	12	\$29,761	\$191,821	\$769	
5	Africa	Addis Ababa	Men's Casual	4	\$67,242	\$118,036	\$2,284	
6	Africa	Addis Ababa	Men's Dress	7	\$76,793	\$136,273	\$2,433	
7	Africa	Addis Ababa	Sandal	10	\$62,819	\$204,284	\$1,861	
8	Africa	Addis Ababa	Slipper	14	\$68,641	\$279,795	\$1,771	
9	Africa	Addis Ababa	Sport Shoe	4	\$1,690	\$16,634	\$79	
10	Africa	Addis Ababa	Women's Casual	2	\$51,541	\$98,641	\$940	
11	Africa	Addis Ababa	Women's Dress	12	\$108,942	\$311,017	\$3,233	
12	Africa	Algiers	Boot	21	\$21,297	\$73,737	\$710	
13	Africa	Algiers	Men's Casual	4	\$63,206	\$100,982	\$2,221	
14								

Freeze Headers + RowHeaders

# AutoFilters for the first four columns

```
ods excel file="C:\MWSUG 2025\ODS Output\Example5.xlsx"
  options ( embedded_titles='yes' zoom='200'
    title_footnote_nobreak='yes'
    frozen_headers='3' frozen_rowheaders='3'
    autofilter='1-4'
    sheet_name='AutoFilters Columns A to D' );
title1 justify=left color=black
'Example 5' color=blue ' AutoFilters for Columns A through D';
options obs=10;
proc print data=sashelp.shoes noobs;
id region subsidiary product;
run;
ods excel close;
options obs=max; /* for next run during this session */
```

# AutoFilters

ExcelExample5.xlsx - Microsoft Excel

File Home Insert Page Layout Formulas Data Review View

Clipboard Font Alignment Number Styles Cells Editing

D4 12

	A	B	C	D	E	F	G	H
1	<b>Example 5 AutoFilters for Columns A through D</b>							
2								
3	<b>Region</b>	<b>Subsidiary</b>	<b>Product</b>	<b>Store</b>	<b>Sales</b>	<b>Inventory</b>	<b>Returns</b>	
4	Africa	Addis Ababa	Boot	12	\$29,761	\$191,821	\$769	
5	Africa	Addis Ababa	Men's Casual	4	\$67,242	\$118,036	\$2,284	
6	Africa	Addis Ababa	Men's Dress	7	\$76,793	\$136,273	\$2,433	
7	Africa	Addis Ababa	Sandal	10	\$62,819	\$204,284	\$1,861	
8	Africa	Addis Ababa	Slipper	14	\$68,641	\$279,795	\$1,771	
9	Africa	Addis Ababa	Sport Shoe	4	\$1,690	\$16,634	\$79	
10	Africa	Addis Ababa	Women's Casual	2	\$51,541	\$98,641	\$940	
11	Africa	Addis Ababa	Women's Dress	12	\$108,942	\$311,017	\$3,233	
12	Africa	Algiers	Boot	21	\$21,297	\$73,737	\$710	
13	Africa	Algiers	Men's Casual	4	\$63,206	\$100,982	\$2,221	
14								

AutoFilters Columns A to D

# Magnifying the AutoFilters

Region 	Subsidiary 	Product 	Stores 
--	--	---	--

# Table and Plot on Separate Worksheets Outside Code


```
ods excel file="C:\MWSUG 2025\ODS Output\Example6.xlsx"
  options ( embedded_titles='yes' zoom='100'
    title_footnote_nobreak='yes'
    sheet_interval='proc' ); /* force separate worksheets */
options obs=10;
ods graphics on / reset=all;
/* turn off any prior graphic settings in this session */
/* here, two code blocks:
  an ods excel options(sheet_name='...'); statement
  and
  the appropriate Plot or Table creation PROC step */
ods excel close;
options obs=max;
```

# Table and Plot on Separate Worksheets Inside Code

```
ods excel options(sheet_name='Plot');
title1 color=black 'Example 6 Plot';
title2 color=blue 'Weight VS Height';
/* no ODS GRAPHICS statement to customize the image */
proc sgplot data=sashelp.class noborder;
scatter x=height y=weight;
xaxis display=(nolabel noline noticks) values=(55 to 70 by 5);
yaxis display=(nolabel noline noticks) values=(80 to 115 by 5);
run;

ods excel options(sheet_name='Table');
title1 color=black 'Example 6 Table';
title2 color=blue 'All Information for 10 Students';
proc print data=sashelp.class noobs;
run;
```

# Plot and Table on Separate Worksheets





# Table and Plot on Same Worksheet Outside Code


```
options obs=10;
ods excel file="C:\MWSUG 2025\ODS Output\Example7.xlsx"
  options ( embedded_titles='yes' title_footnote_nobreak='yes'
    zoom='98' /* adjusted to fit the last row of the table
              when spreadsheet opens on MY computer */
    sheet_interval='none'
    sheet_name='Plot Above Table' );
/* Here, ODS GRAPHICS statement to custom size the image
   so that the plot plus the table below it fill my screen,
   and then PROC steps to create the plot and the table */
ods excel close;
options obs=max;
```

# Table and Plot on Same Worksheet Inside Code

```
ods graphics on / reset=all scale=off
  width=3in; /* sized to fill my screen with table below it */
title1 color=black 'Example 7 Plot';
title2 color=blue 'Weight VS Height';
proc sgplot data=sashelp.class noborder;
scatter x=height y=weight;
xaxis display=(nolabel noline noticks) values=(55 to 70 by 5);
yaxis display=(nolabel noline noticks) values=(80 to 115 by 5);
run;

ods excel options(sheet_name='Table');
title1 color=black 'Example 7 Table';
title2 color=blue 'All Information for 10 Students';
proc print data=sashelp.class noobs;
run;
```

# Table & Plot on Same Worksheet (created at 98% zoom)



# Two Tables Stacked

```
ods excel file="C:\MWSUG 2025\ODS Output\Example8.xlsx"
  options ( embedded_titles='yes' title_footnote_nobreak='yes'
           zoom='98' /* adjusted to show the last row of the bottom table
                    when spreadsheet opens on MY computer */
           sheet_interval='none' sheet_name='Two Tables Stacked' );
title1 color=black 'Example 8'
      color=blue ' Two Tables Stacked In One WorkSheet';
title3 'Girls';
proc print data=sashelp.class noobs;
where Sex='F'; run;
/* Use FOOTNOTE1; here to insert an extra blank row above "Boys" if desired */
title1 'Boys';
proc print data=sashelp.class noobs;
where Sex='M'; run;
ods excel close;
```

# Two Tables Stacked in One WorkSheet

ExcelExample8.xlsx - Microsoft Excel

File Home Insert Page Layout Formulas Data Review View

Clipboard Font Alignment Number Styles Cells Editing

C30

Example 8 Two Tables Stacked In One WorkSheet

**Girls**

Name	Sex	Age	Height	Weight
Alice	F	13	56.5	84.0
Barbara	F	13	65.3	98.0
Carol	F	14	62.8	102.5
Jane	F	12	59.8	84.5
Janet	F	15	62.5	112.5
Joyce	F	11	51.3	50.5
Judy	F	14	64.3	90.0
Louise	F	12	56.3	77.0
Mary	F	15	66.5	112.0

**Boys**

Name	Sex	Age	Height	Weight
Alfred	M	14	69.0	112.5
Henry	M	14	63.5	102.5
James	M	12	57.3	83.0
Jeffrey	M	13	62.5	84.0
John	M	12	59.0	99.5
Philip	M	16	72.0	150.0
Robert	M	12	64.8	128.0
Ronald	M	15	67.0	133.0
Thomas	M	11	57.5	85.0
William	M	15	66.5	112.0

Ready | 123456789112345678912345678 | 98%

## Two Tables Side By Side

Send me an email request for code, if desired.

**OR**

Find it included in a collection of 36 examples,

In my MWSUG 2019 SAS 101 Plus Best Paper

Powerful SAS<sup>®</sup> Output Delivery with ODS Excel

at

<https://www.lexjansen.com/mwsug/2019/SP/MWSUG-2019-SP-072.pdf>

## **For More About ODS Excel**

**The Most Concise Convenient Reference is**

*The ODS Destination for Excel Tip Sheet*

[https://support.sas.com/content/dam/SAS/support/en/product-solutions/base-sas/tip-sheets/ODS\\_Excel\\_Dest\\_tips.pdf](https://support.sas.com/content/dam/SAS/support/en/product-solutions/base-sas/tip-sheets/ODS_Excel_Dest_tips.pdf)

**Out-of-Print, but findable at Amazon:**

**“Exchanging Data From SAS<sup>®</sup> to Excel:**

**The ODS Excel Destination”**

**by William E. Benjamin Jr.**

**A search of the internet or at [lexjanse.com](http://lexjanse.com) for "ODS EXCEL" should turn up other resources.**

**Your questions, comments, and ideas about  
Output Delivery System (ODS) or about communicating with color, graphs,  
plots, charts, or are always welcome:**

LeRoy Bessler PhD

BesslerLeRoy2024@gmail.com

Bessler Consulting and Research

Visual Data Insights™

Strong Smart Systems™

Valparaiso, Indiana, USA

**Visual Data Insights is a trademark of LeRoy Bessler PhD.**

# **Get Your Report to the ODS HTML5 Destination**

LeRoy Bessler PhD

Bessler Consulting and Research

BesslerLeRoy2024@gmail.com

Visual Data Insights™

Strong Smart Systems™

Valparaiso, Indiana, USA

# Presentation Content

**NOTE:** The examples are pasted-in images that were used for my book. If you run the code, your result might look different.

# When It Gets Complicated, But You Want To Present It Visually

- ▶ Next three web pages COULD BE linked to a table/spreadsheet.
- ▶ Code for Figure 14-1 shown here is only snippets.
- ▶ Code for Figure 14-15 is too complicated for slides.
- ▶ Full code for both is in the book, but can be email requested.
- ▶ Code for Figure 14-1 shown here is only snippets.




# Figure 14-1 Key Features of the Code

```
ods graphics on / reset=all scale=off width=5.56in
  imagemap=on /* needed for data tips */
  outputfmt=SVG;


series y=DailyDJ x=Day /
  group=Month /* one plot line for each */
  tip=(Month Day DailyDJ)
  tipformat=(monthnm9. F2. F4.)
  tiplabel=('Month' 'Day' 'Dow Jones Composite Index')
  markers markerattrs=(size=7 symbol=CircleFilled)
  lineattrs=(thickness=3 pattern=Solid);
```

**Figure 14-15. An Alternative to Overlay.** This design could be adapted to the Twelve Months of Trading Days for the Dow Jones Composite Index



**These are Sparse Lines, aka Sparse Line Annotation. See my design practices presentation, or my book, for more information about them. My creation of 1991 before "Spark Lines" by Peter Zelchenko emerged.**

**Figure 14-16.** Markers for Minima & Maxima PLUS Precise Values for them, Median, and Start & End in Maximally Informative Legend



# Common Frame for All ODS HTML5 Examples Used Here

```
ods html5
```

```
/* nogtitle nogfootnote USED to keep them out of the images */
```

```
/* NOGTITLE must be used if using the TITLE LINK= option */
```

```
/* style=WhateverCustomStyleIsUsed (not default HTMLencore) */
```

```
path="C:\YourPreferredFolder"
```

```
body="YourFileName.html" (title='SEE NEXT SLIDE');
```

```
ods graphics on / reset=all width=TBD height=TBD
```

```
scale=off /* I have NEVER found value from scale=on */
```

```
outputfmt=SVG /* default, but best always coded */
```

```
imagemap=on; /* turn on data tips */
```

```
< SAS code goes here >
```

```
ods html5 close;
```

```
ods results on; /* so later processing can open other results in  
this same SAS session, unless results is again turned off */
```

**About** (TITLE='...') suboption of BODY='filename.html'

The text is inserted into invisible METADATA of the HTML file.

The text MIGHT appear at the top left of your web browser window. It MIGHT NOT, but use of the TITLE= options is always harmless.

The screen captures in the slides are clipped at the top. In the book, they are NOT clipped. There, in examples where I did NOT use TITLE=, that part of the screen captured figure shows the default, "SAS Output".

## **Time Saver**

I was a late adopter of ODS Styles.

They are complicated, but a macro can make it easy.

When preparing 327 examples for my book, specifying text characteristics (font, height, Bold or Normal), would have been enormous work.

An ODS Style can define those as defaults for your graph or table, that you CAN override any of them where desired. Send me an email for the macro that I used for this presentation.

## Frequently Used StartUp Code

Entails use of macro `AllTextSetup_LeRBstyle`  
which may do more than affect text characteristics

```
%include "C:\My Macros\AllTextSetup_LeRBstyle.sas";  
%AllTextSetup_LeRBstyle(11,Parent=HTMLencore);
```

```
%AllTextSetup_LeRBstyle(Size,  
Family=Arial,  
Weight=Bold, /* OR Normal */  
TableGrid=ON, /* OR OFF */  
Parent= HTMLencore, /* for ODS PDF, use PRINTER */  
JustifyUserText=CENTER,  
TableGrid=ON); /* OR OFF*/
```

Using Family=Arial Narrow yields SAS Log message:

```
NOTE: STYLE 'ArialNarrow11ptBold_LeRBstyleForHTMLencore' has been  
saved to: SASUSER.TEMPLAT
```

Using Family=Arial,TableGrid=OFF yields SAS Log message:

```
NOTE: STYLE 'Arial11ptBold_LeRBstyleForHTMLencore_NoTableGrid'  
has been saved to: SASUSER.TEMPLAT
```

## Example of Using My ODS Style Macro


To use the macro, you can paste it into the top of your program, above your invocation of the macro.

Instead, if you file it in, e.g., C:\MyMacros, you can start your program with

```
%INCLUDE "C:\MyMacros\AllTextSetupTableNoGridNoPageBrk.sas";
```

In any ODS HTML code that you see here, assume that the %INCLUDE statement was run first. For ways to make the macro available without %INCLUDE, please see the documentation.

# Figure 14-2 Two Graphs Stacked If NO Intervention



- Useless fuzzy light blue background from ODS HTML5 default `STYLE=HTMLencore`
- No value-adding line between plots
- Mixture of font heights and of Bold versus Normal font weight
- The macro can be used to remove the line, make the background white, and make the text black and with uniform characteristics


## Figure 14-2 Code Steps

```
ods html5 path="C:\temp"
style=Arial12ptBold_LeRBstyleForHTMLencore
  body="Fig14-2_TwoGraphsStacked.html"
  (title='Student Average Height By Age');
proc odstext; /* create a common title for both graphs */
p "Average Height By Age" / style=[just=center];
run;
/* setup for both graphs */
ods graphics on / reset=all scale=off width=500px height=200px
  /* Default size in pixels is 640 by 480. */
  outputfmt=SVG
  imagemap;
< PROC SGPLOT steps go here >
Ods html5 close;
```

## Figure 14-2 Code Steps

```
title 'Female Students';
proc sgplot data=sashelp.class (where=(sex EQ 'F'))
  description=' ' noborder; /* NO INNER border */
vbar age / response=height stat=mean datalabel
  displaybaseline=off
  barwidth=0.5 nooutline fillattrs=(color=green);
yaxis display=none;
xaxis display=(nolabel noline noticks);
format height 4.1 age 2.;
run;
/* REPEAT CODE ABOVE with two changes: */
title 'Male Students';
proc sgplot data=sashelp.class (where=(sex EQ 'M'))
  description=' ' noborder;
```

# Figure 14-2 Two Graphs Stacked



## Figure 14-3 Graph Above Table **Code Structure**

```
ods html5 path="C:\temp" nogtitle nogfootnote
  body="Fig14-3_GraphAboveTable.html" (title="...")
  style=Arial11ptBold_LeRBstyleForHTMLencore;
ods graphics on / reset=all scale=off width=500px height=200px
  outputfmt=SVG imagemap=on;
< PROC ODSTEXT step for the title goes here >
< PROC SGPLOT step goes here >
< PROC ODSTEXT step for extra white space goes here >
< PROC PRINT step goes here >
< PROC ODSTEXT step for the footnote goes here >
ods html5 close;
```

## Figure 14-3 Procedure Code Steps (except SGPLOT)

```
proc odstext;
p "Students Whose Name Starts with 'J'" / style=[just=center];
run;

< PROC SGPLOT step goes here >

proc odstext; /* default font_size is too large */
p "Extra White Space" / style=[font_size=2pt color=white];
run;

title justify=center 'Student Information';
proc print data=sashelp.class (where=(name =: 'J')) noobs;
run;


proc odstext;
p "Data Source: SASHELP.CLASS" / style=[just=center];
run;

/* Later, see ODS TEXT= instead of PROC ODSTEXT & P statement */
```

## Figure 14-3 PROC SGPLOT Step

```
title justify=center 'Weight vs Height';
proc sgplot data=sashelp.class (where=(name =: 'J'))
  noborder /* No Inner Border (around the drawing area) */
  description=' '; /* ' ' to prevent useless pop-up of
  "The SGPLOT Procedure" when mouse is moved over the web page */
scatter x=height y=weight /
  tip=(name age height weight) /* tip, etc. for mouseover text */
  tipformat=(auto auto F4.1 F3.)
  tiplabel=('Student' 'Age' 'Height (inches)' 'Weight (pounds)')
  markerattrs=(symbol=CircleFilled color=green);
xaxis display=(noline noticks nolabel) values=(51 to 65 by 2);
yaxis display=(noline noticks nolabel) values=(50 to 120 by 10);
run;
```

# Figure 14-3 Graph Above Table




# Figure 14-4 Graph with Table At Its Right **Code Structure**

```
ods html5 path="C:\temp" nogtitle nogfootnote
  body="Fig14-3_GraphWithTableAtItsRight.html" (title="...")
  style=Arial11ptBold_LeRBstyleForHTMLencore;
ods layout gridded
  columns=2
  column_widths=(240px 240px) column_gutter=20px;
ods region column=1;
ods graphics on / reset=all scale=off width=240px height=200px
  outputfmt=SVG imagemap=on;
< PROC SGPLOT step goes here >
ods region column=2;
< PROC PRINT step goes here >
ods layout end;
ods html5 close;
```

## Figure 14-4 Procedure Code Steps

```
title justify=center 'Weight vs Height';
proc sgplot data=sashelp.class (where=(name =: 'J')) noborder
  description=' ';
scatter x=height y=weight /
  tip=(name age height weight)
  tipformat=(auto auto F4.1 F3.)
  tiplabel=('Student' 'Age' 'Height (inches)' 'Weight (pounds)')
  markerattrs=(symbol=CircleFilled color=green);
xaxis display=(noline noticks nolabel) values=(51 to 63 by 2);
yaxis display=(noline noticks nolabel) values=(50 to 120 by 10);
run;
title justify=center 'Student Information';
proc print data=sashelp.class (where=(name =: 'J')) noobs;
var name height weight; run;
```

# Figure 14-4 Graph With Table At Its Right



# Figure 14-5 Two Graphs Above One Table **Code Structure**

```
ods html5 path="C:\temp" nogtitle nogfootnote
  body="Fig14-5_TwoGraphsAboveOneTable.html" (title="...")
  style=Arial11ptBold_LeRBstyleForHTMLencore;
ods layout gridded columns=2
  column_widths=(210px 210px) column_gutter=20px;
ods graphics on / reset=all scale=off width=210px height=200px
  outputfmt=SVG imagemap=on;
ods region column=1;
< PROC SGPLOT step goes here >
ods region column=2;
< PROC SGPLOT step goes here >
ods layout end;
ods html5 close;
```

# Figure 14-5 Code Structure

```
ods html5 path="C:\temp" nogtitle nogfootnote
  body="Fig14-5_TwoGraphsAboveOneTable.html" (title="...")
  style=Arial11ptBold_LeRBstyleForHTMLencore;
ods layout gridded columns=2
  column_widths=(210px 210px) column_gutter=20px;
ods graphics on / reset=all scale=off width=210px height=200px
  outputfmt=SVG imagemap=on;
< ODS LAYOUT GRIDDED code for two plots goes here >
< ODS LAYOUT GRIDDED code for one table goes here >
ods html5 close;
```

## Figure 14-5 ODS LAYOUT for Left Side Plot & Start

```
ods layout gridded columns=2 /* Define & Start the LAYOUT */
  column_widths=(210px 210px) column_gutter=20px;
ods graphics on / . . .; /* see previous slide */
/* above two statements are for BOTH plots */
ods region column=1;
title justify=center 'Average Height By Age';
proc sgplot data=sashelp.class(where=(name =: 'J'))
  noborder description=' ';
vbar age / response=height stat=mean datalabel
  barwidth=0.5 nooutline fillattrs=(color=green)
  displaybaseline=off;
yaxis display=none;
xaxis display=(noline noticks nolabel);
format height 2.; run;
```


## Figure 14-5 ODS LAYOUT for Right Side Plot & End

```
ods region column=2;
title justify=center 'Average Weight By Age';
proc sgplot data=sashelp.class (where=(name =: 'J'))
  noborder description=' ';
vbar age / response=weight stat=mean datalabel
  barwidth=0.5 nooutline fillattrs=(color=green)
  displaybaseline=off;
yaxis display=none;
xaxis display=(noline noticks nolabel);
format weight 3.;
run;
ods layout end; /* end the first grid (two columns for plots) */
```

## Figure 14-5 ODS LAYOUT (one column) for the Table

```
ods layout gridded columns=1 column_widths=(440px);
ods region column=1;
title justify=center 'Student Information';
footnote "Data Source: SASHELP.CLASS";
proc print data=sashelp.class (where=(name =: 'J')) noobs;
var name sex age height weight;
run;
ods layout end; /* end second grid (one column) */
```

# Figure 14-5 Two Graphs Above One Table



# Figure 14-8 Web Graph with Link **Code Structure**

```
options nocenter; /* UNDO this at the end */
ods html5 path="C:\temp" nogtitle /* essential for LINK= */
  body="Fig14-8_GraphLinkedToTable.html" (title="...")
  style=Arial12ptBold_LeRBstyleForHTMLencore;
ods graphics on / reset=all scale=off noborder
  width=500px height=200px outputfmt=SVG imagemap;
title1 justify=left
  "Average Height By Age of Students Whose Names Start with 'J'";
title2 justify=left
  color=blue underline=1 /* "decorate" the link */
  link="C:\temp\Fig14-8_TableLinkedToGraph.html" /* define it */
  'Go To Table of Information about Students'; /* describe it */
< PROC SGPLOT step goes here >
ods html5 close;
```

# The Essentials for HTML5 Linking

```
ods html5 path="C:\temp" nogtitle /* so that LINK= will work */
  body="Fig14-8_GraphLinkedToTable.html" (title="...")
  style=Arial12ptBold_LeRBstyleForHTMLencore;
title2 justify=left
  color=blue underline=1 /* "decorate" the link
                        Decoration is not a requirement,
                        but it makes the link LOOK like a hyperlink */
link="C:\temp\Fig14-8_TableLinkedToGraph.html" /* define it */
'Go To Table of Information about Students'; /* describe it */
```


# Figure 14-8 & 14-9 PROC SGPLOT Step for the Graph

```
proc sgplot data=sashelp.class (where=(name =: 'J'))
  noborder /* NO inner border (around the drawing area) */
  description=' ' ; /* prevent nuisance
                    of useless pop-up "The SGPLOT Procedure"
                    which is a distraction from USEFUL mouseover text. /
vbar age / response=height stat=mean datalabel
  displaybaseline=off
  barwidth=0.5 nooutline fillattrs=(color=green) ;
yaxis display=none;
xaxis display=(nolabel noline noticks) ;
format height 4.1 age 2.;
run;
```

## Figure 14-8 Code for the Web Table with Link

```
ods html5 path="C:\temp"
  body="Fig14-8_TableLinkedToGraph.html"
  (title="Information about Students Whose Names Start with 'J'")
  style=Arial12ptBold_LeRBstyleForHTMLencore;
title1 justify=left
  "Information about Students Whose Names Start with 'J'";
title2 justify=left color=blue underlin=1
  link="C:\temp\Fig14-8_GraphLinkedToTable.html"
  "Go To Graph of Student Average Height By Age";
proc print data=sashelp.class (where=(name =: 'J')) noobs;
run;
ods html5 close;
options center; /* undo OPTIONS NOCENTER */
```

# Figure 14-8 Web Graph and Web Table InterLinked




## Figure 14-9 Code for the Web Graph with Link

```
options nocenter; /* UNDO this at the end */
ods html5 path="C:\temp" nogtitle /* essential for LINK= */
  body="Fig14-9_GraphLinkedToExcelWorkSheet.html" (title="...")
  style=Arial12ptBold_LeRBstyleForHTMLencore;
ods graphics on / reset=all scale=off noborder
  width=500px height=200px outputfmt=SVG imagemap;
title1 justify=left
  "Average Height By Age of Students Whose Names Start with 'J'";
title2 justify=left color=blue underline=1
  link="C:\temp\Fig14-9_ExcelWorkSheetLinkedToGraph.xlsx"
  'Go To Excel Worksheet of Information about Students';
< PROC SGPLOT step goes here >
ods html5 close;
```


## Figure 14-9 Code for the Spreadsheet with Link

```
ods excel style=AllTextFontArial12ptBold_BlackWhiteTable
  file= "C:\temp\Fig14-9_ExcelWorkSheetLinkedToGraph.xlsx"
  options(embedded_titles='yes'
    title_footnote_nobreak='yes'
    sheet_name='Information about Students');
title1 justify=left
  "Information about Students Whose Names Start with 'J'";
title2 justify=left color=blue underlin=1
  link="C:\temp\Fig14-Fig14-9_GraphLinkedToExcelWorkSheet.html"
  "Go To Graph of Student Average Height By Age";
proc print data=sashelp.class(where=(name =: 'J')) noobs;
run;
ods html5 close;
options center; /* undo OPTIONS NOCENTER */
```

# Figure 14-9 Web Graph and Spreadsheet InterLinked



# Figure 14-12 Web Graph With DrillDown to Six Tables



## Figure 14-12 Code Steps

```
options nocenter; /* Override default. Reset it at The End. */

proc sort data=sashelp.class(keep=Age) out=work.DistinctAges
  nodupkey;
by Age;
run;

/* create a macro variable for each distinct age */
data _null_;
set work.DistinctAges end=LastOne;
call symput('Age' || trim(left(_N_)), trim(left(Age)));
if LastOne;
call symput('HowMany', _N_);
run;
```

## Figure 14-12 Code Steps

```
%macro TableByAge (AgeCount) ;
%do i = 1 %to &AgeCount %by 1;
ods html5 path="C:\temp"
  body="Fig14-12_TableForAge&&Age&i...html"
  (title="Students with Age = &&Age&i")
  style=Arial12ptBold_LeRBstyleForHTMLencore
title1 "Students with Age = &&Age&i";
title2 justify=left color=blue underline=1
  link="C:\temp\Fig14-12_AverageHeightByAge.html"
  'Go To Chart of Average Height By Age';
< PROC PRINT Step goes here >
ods html5 close;
%end;
%mend TableByAge;
```

## Figure 14-12 Code Steps

```
/* PROC PRINT Step for %TableByAge macro: */  
proc print data=sashelp.class (where=(Age EQ &&Age&i)) noobs  
  style(header) = [backgroundcolor=white]; /* replace gray */  
var Name Sex Height Weight;  
run;
```

## Figure 14-12 Code Steps

```
/* create a tabular web page for each age value */
options mprint;
%TableByAge (&HowMany) ;

/* Add a Hyperlink definition variable to the source data set */
data work.ClassWithLinks;
length LinkVar $ 27;
set sashelp.class;
LinkVar = "Fig14-12_TableForAge" || trim(left(Age)) || ".html";
run;
```


## Figure 14-12 Code Steps

```
/* create the web graph with a hot link at each bar */
ods html5 path="C:\temp"
  body="Fig14-12_AverageHeightByAge.html"
  (title="Students Average Height By Age")
  style=AllTextFontArial12ptBold;
ods graphics on / reset=all scale=off width=500px height=200px
  outputfmt=SVG
  imagemap; /* needed for hot links */
< PROC SGPLOT Step goes here >
ods html5 close;
options center; /* undo OPTIONS NOCENTER which would persist
  after this code has run, and potentially cause unwanted
  effects for other code run during the same SAS session */
```


## Figure 14-12 Code Steps

```
/* create the drillable web graph */
title1 justify=center 'Average Height By Age';
title2 justify=center color=blue
      'Click on any bar for table of details';
proc sgplot data=work.ClassWithLinks noborder
      description=' ';
vbar age / response=height stat=mean datalabel
      url=LinkVar /* makes each bar a hot link */
      displaybaseline=off
      barwidth=0.5 nooutline fillattrs=(color=blue);
yaxis display=none;
xaxis display=(noline noticks nolabel);
format height 2.;
run;
```

# Figure 14-12 Web Graph With DrillDown to Six Tables



# Figure 14-16. Markers for Minima & Maxima PLUS Precise Values for them, Median, and Start & End in Maximally Informative Legend



## Figure 14-16 Code Steps

What follows are key coding features.

The code for predecessor processing is in the book,  
and available via email upon request.

## Figure 14-16 Code Steps

What follows are key coding features.

The code for predecessor processing is in the book,  
and available via email upon request.

## Figure 14-16 Code Steps

```
/* requires prior run of Listing 14-0 to create input data */
%include "C:\SharedCode\CommonCodeForFig8_34AndFig14_16.sas";

ods html5 path="C:\temp" style=GraphFontArial18ptBold
  body=
    "Fig14-16_DowByDayEachMonth1990_Overlay_MostInfo_MinMaxMarkers.html"
    (title='Dow Jones Composite Index By Trading Day in 1990');
ods graphics on / reset=all scale=off width=5.56in height=6.4in
  maxlegendarea=30 /* with the default 20,
                    the legend would be omitted */
  imagemap=on /* needed for data tips */
  outputfmt=SVG;
< PROC SGPLOT Step goes here >
ods html5 close;
```

## Figure 14-16 Code Steps

```
/* Graph Creation Step */
title1 justify=center
    "Dow Jones Composite Index By Trading Day in 1990 -
    Range = &MinDJ1990 to &MaxDJ1990, Median = &MedDJ1990";
/* Min, Max, and Median macro variables created in prior step */
title2 justify=center "explain black & white markers here";
proc sgplot data=work.DowJonesByDayIn1990MonthlyMinMax noborder
/* input data created by CommonCodeForFig8_34AndFig14_16.sas */
    description=' ' ; /* ' ' to prevent useless Pop-Up */
< SERIES statement and two SCATTER statements
YAXIS & XAXIS statements
KEYLEGEND statement
FORMAT statement >
run;
```

# Figure 14-16 Code Steps

```
series y=DAILYDJ x=Day / group=Month
    tip=(Month Day DailyDJ)
    tipformat=(monthnm9. F2. F4.)
    tiplabel=('Month' 'Day' 'Dow Jones Composite Index')
    markers markerattrs=(size=7 symbol=CircleFilled)
    lineattrs=(thickness=3 pattern=Solid);
scatter x=Day y=MinDJ /
    FilledOutlinedMarkers
    markerattrs=(symbol=SquareFilled size=7px)
    markerfillattrs=(color=white)
    markeroutlineattrs=(color=black thickness=1px);
scatter x=Day y=MaxDJ /
    markerattrs=(symbol=SquareFilled color=black
    size=9px); /* 9px match size of 7px outlined white markers */
```

## Figure 14-16 Code Steps

```
yaxis display=(nolabel noticks noline) grid minorgrid
minorcount=1
    values=(830 to 1080 by 20);
xaxis display=(nolabel noticks noline) grid fitpolicy=none
    values=(1 TO 31 BY 1);
keylegend / title='' noborder across=1;
format DailyDJ 4. Month MonthStats.;
/* MonthStats format created by CommonCodeForFig8_34AndFig9_16
    delivers the twelve labels in the legend. */
```

## For More About ODS HTML5

A search of the internet turned up three resources:

▶ A reference to Chapter 14 in my book *Visual Data Insights Using SAS ODS Graphics: A Guide to Communication-Effective Data Visualization*

▶ SAS documentation (as of 28 April 2025) for the ODS HTML5 Statement (with internal links that might be of interest to you):  
[https://documentation.sas.com/doc/en/pgmsascdc/9.4\\_3.5/odsug/p0hcv8gpxqebnplis52we2enltx.htm](https://documentation.sas.com/doc/en/pgmsascdc/9.4_3.5/odsug/p0hcv8gpxqebnplis52we2enltx.htm)

Only one SAS users conference paper from 2018 (for which I cannot speak as to the currency of its information):

<https://support.sas.com/resources/papers/proceedings18/2670-2018.pdf>

**Your questions, comments, and ideas about  
Output Delivery System (ODS) or about communicating with color, graphs,  
plots, charts, or are always welcome:**

LeRoy Bessler PhD

BesslerLeRoy2024@gmail.com

Bessler Consulting and Research

Visual Data Insights™

Strong Smart Systems™

Valparaiso, Indiana, USA

**Visual Data Insights is a trademark of LeRoy Bessler PhD.**

# **Get Your Reports to the PDF Destination**

LeRoy Bessler PhD

Bessler Consulting and Research

BesslerLeRoy2024@gmail.com

Visual Data Insights™

Strong Smart Systems™

Valparaiso, Indiana, USA

# Presentation Content

**NOTE:** The examples are pasted-in screen captures.  
Some of the open PDF pages were zoomed before capture.  
Your initially opened result from a code run may look different.

# My Standard PDF Code Block Structure

```
/* Preliminary: */  
options nodate nonumber nocenter;  
ods results off; /* nothing to Output window */  
ods _all_ close;  
title; footnote; /* erase leftovers from prior code  
                  during the current session */  
  
/* SOMETIMES used:  
ods escapechar='^'; for in-line formatting  
of text from ODS TEXT or ODS PDF TEST= statement */
```

# My Standard PDF Code Block Structure

```
/* The Essential: */  
ods pdf  
  notoc  
  style=SomeCustomStyleName /* STYLE= is optional */  
  file=" &Path.\FileName.pdf";  
  < application code here >  
ods pdf close;
```

NOTE: To save space, ODS PDF CLOSE; will sometimes be omitted in code shown on slides.

# Why NOTOC?

Bookmarks and the Table Of Contents Page options create a situation where clicking on the bookmark or the table of contents entry DOES NOT take the viewer to the TOP of the target. There is no prospect of a cure for this.

```
ods escapechar='^';
```

You will see how the escape character is used for “in-line formatting” (To Be Defined and Demonstrated).

You may use a different character, if you wish.

If no in-line formatting is done,  
this statement is unnecessary,  
but its use at StartUp does no harm.

**The default page orientation in ODS PDF is PORTRAIT.**

**For LANDSCAPE, use  
OPTIONS ORIENTATION=LANDSCAPE;**

**The OPTIONS statement may be either before or after the start of the ODS code block—i.e., before or after ODS PDF FILE=**

**The ORIENTATION option (default or specified) remains in effect for the duration of the SAS session, unless overridden with another use of  
OPTIONS ORIENTATION=**

## Frequently Used StartUp Code

Entails use of macro `AllTextSetup_LeRBstyle`  
which may do more than affect text characteristics

```
%include "C:\My Macros\AllTextSetup_LeRBstyle.sas";  
%AllTextSetup_LeRBstyle(11);
```

```
%AllTextSetup_LeRBstyle(Size,  
Family=Arial,  
Weight=Bold, /* OR Normal */  
TableGrid=ON, /* OR OFF */  
Parent=PRINTER, /* for ODS HTML5, use HTMLencore */  
JustifyUserText=CENTER,  
TableGrid=ON); /* OR OFF*/
```

Using Family=Arial Narrow yields SAS Log message:

```
NOTE: STYLE 'ArialNarrow11ptBold_LeRBstyleForPRINTER' has been  
saved to: SASUSER.TEMPLAT
```

Using Family=Arial,TableGrid=OFF yields SAS Log message:

```
NOTE: STYLE 'Arial11ptBold_LeRBstyleForPRINTER_NoTableGrid' has  
been saved to: SASUSER.TEMPLAT
```

# Default Table

```
options nodate nonumber nocenter;
ods results off;
ods _all_ close;
title; footnote;
ods pdf notoc /* NO STYLE= */
  file="C:\temp\Fig17-0_TableDefaultStyleForPDF.pdf";
title1 "Students with Age=13";
proc print data=sashelp.class noobs;
where Age EQ 13;
run;
ods pdf close;
```

# Default ODS PDF Style Table - Magnified 300%

Fig17-0\_TableDefaultStyleForPDF.pdf - Adobe Acrobat Reader (64-bit)

File Edit View Sign Window Help

Home Tools Fig17-0\_TableDefau... x

1 / 1 300%

## Students with Age=13

Name	Sex	Age	Height	Weight
Alice	F	13	56.5	84
Barbara	F	13	65.3	98
Jeffrey	M	13	62.5	84

```
%include
"C:\SharedCode\AllTextSetup_LeRBstyle.sas";
%AllTextSetup_LeRBstyle(11);
options nodate nonumber nocenter;
ods results off;
ods _all_ close;
title; footnote;
ods pdf notoc
  style=Arial11ptBold_LeRBstyleForPRINTER
  file="C:\temp\Fig17-1_LeRBstyle11ptBoldTableWithGrid.pdf";
title1 "Students with Age=13";
proc print data=sashelp.class noobs;
where Age EQ 13;
run;
ods pdf close;
```

# Arial 11pt Bold LeRBstyle Table With Grid

Fig17-1\_LeRBstyle11ptBoldTableWithGrid.pdf - Adobe Acrobat Reader (64-bit)

File Edit View Sign Window Help

Home Tools Fig17-1\_LeRBstyle1... x

1 / 1 300%

## Students with Age=13

<b>Name</b>	<b>Sex</b>	<b>Age</b>	<b>Height</b>	<b>Weight</b>
<b>Alice</b>	<b>F</b>	<b>13</b>	<b>56.5</b>	<b>84</b>
<b>Barbara</b>	<b>F</b>	<b>13</b>	<b>65.3</b>	<b>98</b>
<b>Jeffrey</b>	<b>M</b>	<b>13</b>	<b>62.5</b>	<b>84</b>

```
%include
  "C:\SharedCode\AllTextSetup_LeRBstyle.sas";
%AllTextSetup_LeRBstyle(11,TableGrid=OFF);
options nodate nonumber nocenter;
ods results off;
ods _all_ close;
title; footnote;
ods pdf notoc
  style=Aria11ptBold_LeRBstyleForPRINTER_NoTableGrid
file="C:\temp\Fig17-2_LeRBstyle11ptBoldTableWithoutGrid.pdf";
title1 "Students with Age=13";
proc print data=sashelp.class noobs;
where Age EQ 13;
run;
ods pdf close;
```

# Arial 11pt Bold LeRBstyle Table Without Grid

Fig17-2\_LeRBstyle11ptBoldTableWithoutGrid.pdf - Adobe Acrobat Reader (64-bit)

File Edit View Sign Window Help

Home Tools Fig17-2\_LeRBstyle1... x

1 / 1 300%

## Students with Age=13

Name	Sex	Age	Height	Weight
Alice	F	13	56.5	84
Barbara	F	13	65.3	98
Jeffrey	M	13	62.5	84

# Two Tables Stacked

```
options nodate nonumber nocenter;
ods results off;
ods _all_ close;
title; footnote;
ods pdf notoc
  StartPage=never /* Assure All Output on the Same Page */
  file="C:\temp\Fig17-3A_TwoTablesStacked_2ndTitleMissing.pdf";
title1 "Height of Students with Age=13";
proc print data=sashelp.class noobs;
where Age EQ 13;
var Name Height;
run;
title1 "Weight of Students with Age=13"; /* DOES NOT APPEAR */
proc print data=sashelp.class noobs;
where Age EQ 13;
var Name Weight;
run;
ods pdf close;
```

# Two Tables Stacked – ODS PDF StartPage=NEVER – TITLE2 missing

Fig17-3A\_TwoTablesStacked\_2ndTitleMissing.pdf - Adobe Acrobat Reader (64-bit)

File Edit View Sign Window Help

Home Tools Fig17-3A\_TwoTable... x

1 / 1 150%

## Height of Students with Age=13

Name	Height
Alice	56.5
Barbara	65.3
Jeffrey	62.5

Name	Weight
Alice	84
Barbara	98
Jeffrey	84

```
options nodate nonumber nocenter;
ods results off; ods _all_ close; title; footnote;
ods pdf notoc
  file=
"C:\temp\Fig17-3B_TwoTablesStacked_UsingTwoRowGriddedLayout.pdf";
ods layout gridded rows=2;
ods region row=1;
title1 "Height of Students with Age=13";
proc print data=sashelp.class noobs;
where Age EQ 13;
var Name Height; run;
ods region row=2;
title1 "Weight of Students with Age=13";
proc print data=sashelp.class noobs;
where Age EQ 13;
var Name Weight; run;
ods layout end;
ods pdf close;
```

# Two Tables Stacked – ODS LAYOUT GRIDDED ROWS=2

Fig17-3B\_TwoTablesStacked\_UsingTwoRowGriddedLayout.pdf - Adobe Acrobat Reader (64-bit)

File Edit View Sign Window Help

Home Tools Fig17-3B\_TwoTable... x

1 / 1 150%

## Height of Students with Age=13

Name	Height
Alice	56.5
Barbara	65.3
Jeffrey	62.5

## Weight of Students with Age=13

Name	Weight
Alice	84
Barbara	98
Jeffrey	84

# Two Tables Side By Side

```
options nodate nonumber nocenter;
ods results off;
ods _all_ close;
title; footnote;
ods pdf notoc
    columns=2 /* VERY EASY, but with INCOMPLETE Second Column */
file="C:\temp\Fig17-4A_TwoTablesSideBySide_2ndTitleMissing.pdf";
title1 "Height of Students with Age=13";
proc print data=sashelp.class noobs;
where Age EQ 13;
var Name Height;
run;
title1 "Weight of Students with Age=13"; /* DOES NOT APPEAR */
proc print data=sashelp.class noobs;
where Age EQ 13;
var Name Weight;
run;
ods pdf close;
```

# Two Tables SideBySide - ODS PDF COLUMNS=2 - TITLE2 missing

Fig17-4A\_TwoTablesSideBySide\_2ndTitleMissing.pdf - Adobe Acrobat Reader (64-bit)

File Edit View Sign Window Help

Home Tools Fig17-4A\_TwoTable... x

1 / 1 200%

## Height of Students with Age=13

Name	Height
Alice	56.5
Barbara	65.3
Jeffrey	62.5

Name	Weight
Alice	84
Barbara	98
Jeffrey	84

```
options nodate nonumber nocenter;
ods results off; ods _all_ close; title; footnote;
ods pdf notoc
  file="C:\temp\
  Fig17-4B_TwoTablesSideBySide_UsingTwoColumnGriddedLayout.pdf";
ods layout gridded columns=2;
ods region column=1;
title1 "Height of Students with Age=13";
proc print data=sashelp.class noobs;
where Age EQ 13;
var Name Height; run;
ods region column=2;
title1 "Weight of Students with Age=13";
proc print data=sashelp.class noobs;
where Age EQ 13;
var Name Weight; run;
ods layout end;
ods pdf close;
```

# Two Tables SideBySide - ODS LAYOUT GRIDDED COLUMNS=2

Fig17-4B\_TwoTablesSideBySide\_UsingTwoColumnGriddedLayout.pdf - Adobe Acrobat Reader (64-bit)

File Edit View Sign Window Help

Home Tools Fig17-4B\_TwoTable... x

1 / 1 200%

## Height of Students with Age=13

Name	Height
Alice	56.5
Barbara	65.3
Jeffrey	62.5

## Weight of Students with Age=13

Name	Weight
Alice	84
Barbara	98
Jeffrey	84

ods pdf notoc **StartPage=never**

**Without StartPage=never,**  
every PROC step in the ODS PDF code block  
triggers a page break.

# Two Graphs Stacked

```
options nodate nonumber center;
ods results off;
ods _all_ close;
title; footnote;
ods pdf notoc style=Arial11ptBold_LeRBstyleForPRINTER
  startpage=never /* NO new page for second graph */
  file="C:\temp\Fig16-2_TwoGraphsStacked.pdf";
/* create a common title for both graphs */
ods text="Average Height By Age";
/* setup for both graphs */
ods graphics on / reset=all scale=off outputfmt=SVG
  width=500px height=200px;
  /* Default size 640px by 480px */
/* two PROC SGPLOT steps here */
ods pdf close;
```

```
/* Graph Creation PROC SGPLOT steps */
title 'Female Students';
proc sgplot data=sashelp.class(where=(sex EQ 'F'))
  description=' ' noborder; /* NO INNER border */
vbar age / response=height stat=mean datalabel
  displaybaseline=off
  barwidth=0.5 nooutline fillattrs=(color=green);
yaxis display=none;
xaxis display=(nolabel noline noticks);
format height 4.1 age 2.;
run;
/* REPEAT CODE ABOVE with two changes: */
title 'Male Students';
proc sgplot data=sashelp.class(where=(sex EQ 'M'))
  description=' ' noborder;
```

# Two Graphs Stacked



Fig16-2\_TwoGraphsStacked.pdf - Adobe Acrobat Reader (64-bit)

File Edit View Sign Window Help

Home Tools Fig16-2\_TwoGraph... x

1 / 1 100%

### Average Height By Age



**Graph Above Table**

```
options nodate nonumber center;
ods results off; ods _all_ close; title; footnote;
ods escapechar='^'; /* for inline formatting of text */
ods pdf notoc style=Arial11ptBold_LeRBstyleForPRINTER
  startpage=never file="C:\temp\Fig16-3_GraphAboveTable.pdf";
/* create a common title */
ods text="Students Whose Name Starts with 'J'";
ods text="^S={font_size=8pt color=white}White Space";
ods graphics on / reset=all scale=off outputfmt=SVG
  width=500px height=200px;
/* PROC SGPLOT, White Space insert step, PROC PRINT here */
ods text="^S={font_size=12pt color=white}White Space";
ods text="Data Source: SASHELP.CLASS"; /* common footnote */
ods pdf close;
```

```
/* create the content with white spaces & common footnote */
title justify=center 'Weight vs Height';
proc sgplot data=sashelp.class(where=(name =: 'J'))
  noborder; /* No Inner Border (around the drawing area) */
scatter x=height y=weight /
  markerattrs=(symbol=CircleFilled color=green);
xaxis display=(noline noticks nolabel) values=(51 to 65 by 2);
yaxis display=(noline noticks nolabel) values=(50 to 120 by 10);
run;

ods text="^S={font_size=8pt color=white}White Space";
title justify=center 'Student Information';
proc print data=sashelp.class(where=(name =: 'J')) noobs;
run;

ods text="^S={font_size=12pt color=white}White Space";
ods text="Data Source: SASHELP.CLASS"; /* common footnote */
```

# Graph Above Table


Fig16-3\_GraphAboveTable.pdf - Adobe Acrobat Reader (64-bit)

File Edit View Sign Window Help

Home Tools Fig16-3\_GraphAbo... x

1 / 1 100%

## Students Whose Name Starts with 'J'



Name	Sex	Age	Height	Weight
James	M	12	57.3	83.0
Jane	F	12	59.8	84.5
Janet	F	15	62.5	112.5
Jeffrey	M	13	62.5	84.0
John	M	12	59.0	99.5
Joyce	F	11	51.3	50.5
Judy	F	14	64.3	90.0

# Graph With Table At Its Right

```
ods pdf notoc style=Arial11ptBold_LeRBstyleForPRINTER
startpage=no file="C:\temp\Fig16-4_GraphWithTableAtItsRight.pdf";
ods text="Students Whose Name Starts with 'J'";
ods text="^S={font_size=8pt color=white}White Space";
ods layout gridded columns=2 column_widths=(500px 240px);
ods region column=1;
ods graphics on / reset=all scale=off outputfmt=SVG width=500px
    height=368px; /* chosen so that the table is slightly taller */
    /* white space drawn immediately below tallest content above */
/* PROC SGPLOT step */
ods region column=2;
/* PROC PRINT step */
ods layout end;
/* white space is drawn above common footnote insert */
ods html5 close;
```

```
/* create the content with white space & footnote below it */
ods region column=1; title justify=center 'Weight vs Height';
proc sgplot data=sashelp.class(where=(name =: 'J')) noborder;
scatter x=height y=weight /
    markerattrs=(symbol=CircleFilled color=green);
xaxis display=(noline noticks nolabel) values=(51 to 65 by 2);
yaxis display=(noline noticks nolabel) values=(50 to 120 by 10);
run;

ods region column=2;
title justify=center 'Student Information';
proc print data=sashelp.class(where=(name =: 'J')) noobs;
var Name Weight Height; run;

ods layout end;

ods text="^S={font_size=20pt color=white}White Space";
ods text="Data Source: SASHELP.CLASS";
```

# Graph With Table At Its Right


Fig16-4\_GraphWithTableAtItsRight.pdf - Adobe Acrobat Reader (64-bit)

File Edit View Sign Window Help

Home Tools Fig16-4\_GraphWit... x

1 / 1 135%

## Students Whose Name Starts with 'J'



## Student Information

Name	Weight	Height
James	83.0	57.3
Jane	84.5	59.8
Janet	112.5	62.5
Jeffrey	84.0	62.5
John	99.5	59.0
Joyce	50.5	51.3
Judy	90.0	64.3

Data Source: SASHELP.CLASS

```
/* Differences from code for Table With Grid */
```

```
%include "C:\SharedCode\AllTextSetup_LeRBstyle.sas";
```

```
%AllTextSetup_LeRBstyle(11,TableGrid=OFF)
```

```
ods pdf notoc style=Arial11ptBold_LeRBstyleForPRINTER_NoTableGrid
```

```
file="C:\temp\Fig16-4_GraphWithTableAtItsRight_NoTableGrid.pdf";
```

```
ods graphics on / reset=all scale=off outputfmt=SVG width=500px
```

```
height=350px; /* Make the graph shorter because the table  
without a grid is shorter than the graph with 368px a height, and  
the white space below the pair would overwrite the bottom of that  
taller graph. */
```

# Graph With Table At Its Right


Fig16-4\_GraphWithTableAtItsRight\_NoTableGrid.pdf - Adobe Acrobat Reader (64-bit)

File Edit View Sign Window Help

Home Tools Fig16-4\_GraphWit... x

1 / 1 135%

## Students Whose Name Starts with 'J'



## Student Information

Name	Weight	Height
James	83.0	57.3
Jane	84.5	59.8
Janet	112.5	62.5
Jeffrey	84.0	62.5
John	99.5	59.0
Joyce	50.5	51.3
Judy	90.0	64.3

Data Source: SASHELP.CLASS

**The table without a grid is not an inherently better design than the table with a grid.**

**The benefit of the demonstration is to draw attention to the fact that ODS TEXT= starts writing its output below the LAST object written to the output area.**

Two Graphs Above One Table (uses nested layouts)

```
options nodate nonumber center;
ods results off; ods _all_ close; title; footnote;
ods escapechar='^';
ods pdf notoc style=Arial11ptBold_LeRBstyleForPRINTER
  file="C:\temp\Fig16-5_TwoGraphsAboveOneTable.pdf";
ods text="Students Whose Name Starts with 'J'";
ods text="^S={font_size=8pt color=white}White Space";
ods layout gridded rows=2;
/* code for region definitions and other content goes here */
ods layout end;
ods text="^S={font_size=20pt color=white}White Space";
ods text="Data Source: SASHELP.CLASS";
ods pdf close;
```

```
ods region row=1;

ods layout gridded columns=2
  column_widths=(3in 3in) column_gutter=0.5in;
ods graphics on / reset=all scale=off width=3in
  outputfmt=SVG;
ods region column=1;
/* PROC SGPLOT step for Average Height By Age goes here */
ods region column=2;
/* PROC SGPLOT step for Average Weight By Age goes here */
ods layout end; /* NOT really the end of the complete layout,
  but a signal that the gridded layout inside row=1 has ended. */
```

```
ods region column=1; /* INSIDE the ROW=1 part of the layout */
title 'Average Height By Age';
proc sgplot data=sashelp.class(where=(name =: 'J')) noborder;
vbar age / response=height stat=mean datalabel barwidth=0.5
    displaybaseline=off nooutline fillattrs=(color=green);
yaxis display=none; xaxis display=(nolabel noline noticks);
format height 2.; run;
```

```
ods region column=2; /* INSIDE the ROW=1 part of the layout */
title 'Average Weight By Age';
proc sgplot data=sashelp.class(where=(name =: 'J')) noborder;
vbar age / response=weight stat=mean datalabel barwidth=0.5
    displaybaseline=off nooutline fillattrs=(color=green);
yaxis display=none; xaxis display=(nolabel noline noticks);
format weight 3.; run;
```

```
ods region row=2;
```

```
title1 height=6pt color=white 'White Space';
```

```
title2 justify=center 'Student Information';
```

```
proc print data=sashelp.class (where=(name =: 'J')) noobs;
```

```
format height 2.;
```

```
format weight 3.;
```

```
run;
```

```
ods layout end; /* END to the two-row layout */
```

# Two Graphs Above One Table



Fig16-5\_TwoGraphsAboveOneTable.pdf - Adobe Acrobat Reader (64-bit)

File Edit View Sign Window Help

Home Tools Fig16-4\_GraphWit... Fig16-5\_TwoGraph... x

1 / 1 75%

Students Whose Name Starts with 'J'



Student Information

Name	Sex	Age	Height	Weight
James	M	12	57	83
Jane	F	12	60	85
Janet	F	15	63	113
Jeffrey	M	13	63	84
John	M	12	59	100
Joyce	F	11	51	51
Judy	F	14	64	90

Data Source: SASHELP.CLASS

# Links

# Links Between Two PDF files

```
options nodate nonumber center;  
ods results off; ods _all_ close; title; footnote;  
ods pdf notoc  
file="C:\temp\Fig18-1_ScatterPlotLinkedToTable.pdf"  
style=Arial11ptBold_LeRBstyleForPRINTER  
dpi=600 /* prevent underline anomaly */  
nogtitle; /* so that the LINK= option works */  
ods graphics on / reset=all scale=off width=3in height=3in;
```

```
title1 "Weight vs Height in SASHELP.CLASS";
title2 "for Students Whose Name Starts with 'J'";
title3 color=blue underlin=1
      link="Fig18-1_TableLinkedToScatterPlot.pdf"
      "Click for Table of All Student Information";
proc sgplot data=sashelp.class noborder;
where Name =: 'J';
scatter y=weight x=height;
yaxis display=(noline noticks nolabel) values=(50 to 150 by 10);
xaxis display=(noline noticks nolabel) values=(51 to 65 by 1) fitpolicy=stagger;
run;
ods pdf close;
```

```
options nodate nonumber center;
ods results off; ods _all_ close; title; footnote;
ods pdf notoc
  file="C:\temp\Fig18-1_TableLinkedToScatterPlot.pdf"
  dpi=600 /* prevent underline anomaly */
  style=AllTextFontArial11ptBold_ODSpdf;
title1 "Information in SASHELP.CLASS";
title2 "for Students Whose Name Starts with 'J'";
title3 color=blue underlin=1
  link="Fig18_ScatterPlotLinkedToTable.pdf"
  "Click for Plot of Student Weight vs Height";
proc print data=sashelp.class noobs;
where Name =: 'J';
run;
ods pdf close;
/* links can also be nonPDF destinations */
```

# Scatter Plot Linked To Table


Fig18-1\_ScatterPlotLinkedToTable.pdf - Adobe Acrobat Reader (64-bit)

File Edit View Sign Window Help

Home Tools Fig18-1\_ScatterPlot... x

1 / 1 135%

**Weight vs Height in SASHELP.CLASS**  
for Students Whose Name Starts with 'J'  
[Click for Table of All Student Information](#)



# Table Linked To Scatter Plot

Fig18-1\_TableLinkedToScatterPlot.pdf - Adobe Acrobat Reader (64-bit)

File Edit View Sign Window Help

Home Tools Fig18-1\_TableLinke... x

1 / 1 135%

**Information in SASHELP.CLASS**  
**for Students Whose Name Starts with 'J'**  
[Click for Plot of Student Weight vs Height](#)

Name	Sex	Age	Height	Weight
James	M	12	57.3	83.0
Jane	F	12	59.8	84.5
Janet	F	15	62.5	112.5
Jeffrey	M	13	62.5	84.0
John	M	12	59.0	99.5
Joyce	F	11	51.3	50.5
Judy	F	14	64.3	90.0

# Links Between Outputs Within a PDF File

```
ods pdf notoc
file="C:\temp\Fig18-2_ScatterPlotLinkedToTableInsideThisPDFfile.pdf"
style=Arial11ptBold_LeRBstyleForPRINTER
dpi=600
nogtitle; /* put graph titles outside its image so that the links work */
ods graphics on / reset=all scale=off width=3in height=3in;
ods pdf anchor='Plot';
/* code to create Plot with link to Table anchor goes here */
ods startpage=yes; /* specify no
if you improbably want interlinked plot and table on the same page */
ods pdf anchor='Table';
/* code to create Table with link to Plot anchor goes here */
ods pdf close;
```

```
ods pdf anchor='Plot';
title1 color=white 'White Space'; /* so that return to Plot anchor exposes Titles 3,4,5 */
title2 color=white 'White Space'; /* so that return to Plot anchor exposes Titles 3,4,5 */
title3 "Weight vs Height in SASHELP.CLASS";
title4 "for Students Whose Name Starts with 'J'";
title5 color=blue underlin=1
      "^S={URL='#Table'}Click for Table of All Student Information";
proc sgplot data=sashelp.class noborder;
where Name =: 'J';
scatter y=weight x=height;
yaxis display=(noline noticks nolabel) values=(50 to 150 by 10);
xaxis display=(noline noticks nolabel) values=(51 to 65 by 1) fitpolicy=stagger;
run;
```

```
ods startpage=yes; /* specify no
if you improbably want plot and table on the same page */
ods pdf anchor='Table';
title1 color=white 'White Space'; /* so that return to Plot anchor exposes Titles 3,4,5 */
title2 color=white 'White Space'; /* so that return to Plot anchor exposes Titles 3,4,5 */
title3 "Information in SASHELP.CLASS";
title4 "for Students Whose Name Starts with 'J'";
title5 color=blue underlin=1
"^S={URL='#Plot'}Click for Plot of Student Weight vs Height";
proc print data=sashelp.class noobs;
where Name =: 'J';
run;

ods pdf close;
```

# Scatter Plot with Link to Table on a Separate Page of the PDF File

## Zoomed to 100% after opening the file

Fig18-2\_ScatterPlotLinkedToTableInsideThisPDFfile.pdf - Adobe Acrobat Reader (64-bit)


File Edit View Sign Window Help

Home Tools Fig18-2\_ScatterPlot... x

1 / 2

100%

Weight vs Height in SASHELP.CLASS  
for Students Whose Name Starts with 'J'  
[Click for Table of All Student Information](#)



Height	Weight
51	50
57	83
58	85
59	100
60	85
62	85
63	115
64	90

# Table with Link to Scatter Plot on a Separate Page of the PDF File Opened at 100% Zoom after clicking the link in the scatter plot

Fig18-2\_ScatterPlotLinkedToTableInsideThisPDFfile.pdf - Adobe Acrobat Reader (64-bit)

File Edit View Sign Window Help

Home Tools Fig18-2\_ScatterPlot... x

2 / 2 100%

Information in SASHELP.CLASS  
for Students Whose Name Starts with 'J'  
[Click for Plot of Student Weight vs Height](#)

Name	Sex	Age	Height	Weight
James	M	12	57.3	83.0
Jane	F	12	59.8	84.5
Janet	F	15	62.5	112.5
Jeffrey	M	13	62.5	84.0
John	M	12	59.0	99.5
Joyce	F	11	51.3	50.5
Judy	F	14	64.3	90.0

**Demonstrate the Benefit of White Spaces Above the Real Titles**

```
ods pdf notoc
```

```
file="C:\temp\Fig18-3_NoWhiteSpaceAboveTitlesOf  
ScatterPlotAndTableInsideThisPDFfileScatterPlotLinkedToTableInsideThisPDFfile.pdf"  
style=Arial11ptBold_LeRBstyleForPRINTER  
dpi=600
```

```
nogtitle; /* put graph titles outside its image so that the links work */  
ods graphics on / reset=all scale=off width=3in height=3in;
```

```
ods pdf anchor='Plot';
```

```
/* code to create Plot with link to Table anchor goes here */
```

```
ods startpage=yes;
```

```
ods pdf anchor='Table';
```

```
/* code to create Table with link to Plot anchor goes here */
```

```
ods pdf close;
```

```
ods pdf anchor='Plot';
title1 "Weight vs Height in SASHELP.CLASS";
title2 "for Students Whose Name Starts with 'J'";
title3 color=blue underlin=1
      "^S={URL='#Table'}Click for Table of All Student Information";
proc sgplot data=sashelp.class noborder;
where Name =: 'J';
scatter y=weight x=height;
yaxis display=(noline noticks nolabel) values=(50 to 150 by 10);
xaxis display=(noline noticks nolabel) values=(51 to 65 by 1) fitpolicy=stagger;
run;
```

```
ods startpage=yes;
ods pdf anchor='Table';
title1 "Information in SASHELP.CLASS";
title2 "for Students Whose Name Starts with 'J'";
title3 color=blue underlin=1
"^S={URL='#Plot'}Click for Plot of Student Weight vs Height";
proc print data=sashelp.class noobs;
where Name =: 'J';
run;

ods pdf close;
```

# Scatter Plot with Link to Table on a Separate Page of the PDF File

## Zoomed to 100% after opening the file


Fig18-3\_NoWhiteSpaceAboveTitlesOfScatterPlotAndTableInsideThisPDFfile.pdf - Adobe Acrobat Reader (64-bit)

File Edit View Sign Window Help

Home Tools Fig18-3\_NoWhiteS... x

1 / 2 100%

**Weight vs Height in SASHELP.CLASS  
for Students Whose Name Starts with 'J'**  
[Click for Table of All Student Information](#)



The scatter plot displays the relationship between height (x-axis) and weight (y-axis) for students whose names start with 'J'. The x-axis ranges from 51 to 65, and the y-axis ranges from 50 to 150. There are 10 data points plotted, showing a clear upward trend.

Height	Weight
51	50
57	85
58	85
59	100
60	85
61	85
62	85
63	115
64	90
65	90

# Table with Link to Scatter Plot on a Separate Page of the PDF File (obscured TITLE1) Arrived here by clicking link in scatter plot

Fig18-3\_NoWhiteSpaceAboveTitlesOfScatterPlotAndTableInsideThisPDFfile.pdf - Adobe Acrobat Reader (64-bit)

File Edit View Sign Window Help

Home Tools Fig18-3\_NoWhiteS... x

2 / 2

100%

for Students whose name starts with J  
[Click for Plot of Student Weight vs Height](#)

Name	Sex	Age	Height	Weight
James	M	12	57.3	83.0
Jane	F	12	59.8	84.5
Janet	F	15	62.5	112.5
Jeffrey	M	13	62.5	84.0
John	M	12	59.0	99.5
Joyce	F	11	51.3	50.5
Judy	F	14	64.3	90.0

# Scatter Plot with Link to Table on a Separate Page of the PDF File (obscured TITLE1) Arrived here by clicking link in the Table

Fig18-3\_NoWhiteSpaceAboveTitlesOfScatterPlotAndTableInsideThisPDFfile.pdf - Adobe Acrobat Reader (64-bit)

File Edit View Sign Window Help

Home Tools Fig18-3\_NoWhiteS... x

1 / 2

100%

for Students whose Name Starts with J  
[Click for Table of All Student Information](#)

Student ID	Value
51	50
57	85
59	100
60	85
61	85
62	85
63	115
64	90

# ODS PDF Compression

```
ods pdf file="C:\temp\CompressedPDFfile.pdf"  
  compress=N . . . ;
```

N is integer in range 0 to 9, where 0 is for No Compression, 9 is for Maximum Compression,

I found that 1 delivers the maximum effect, and other values deliver smaller increments

# Information for the PDF document Properties tab

```
ods pdf file="C:\temp\PDFfileWithCustomizedProperties.pdf"  
  author = 'SAS Author'  
  title = 'SAS Paper'  
  subject = 'SAS Subject'  
  keywords = 'SAS ODS PDF output';  
...  
ods pdf close;
```

For More Information About ODS PDF ([list on next page](#))

**For Latest Doc for the ODS PDF Statement As Of April 28, 2025:**

**[https://documentation.sas.com/doc/en/pgmsascdc/9.4\\_3.5/odsug/n0mc4eolqoned0n16oy88mpj0e4g.htm](https://documentation.sas.com/doc/en/pgmsascdc/9.4_3.5/odsug/n0mc4eolqoned0n16oy88mpj0e4g.htm)**

**For ODS PDF Tip Sheet:**

**<https://support.sas.com/content/dam/SAS/support/en/products-solutions/base-sas/tip-sheets/PDF-tips.pdf>**

**For ODS LIST AND TEXT BLOCK Tip Sheet:**

**[https://support.sas.com/content/dam/SAS/support/en/products-solutions/base-sas/tip-sheets/Tipsheet\\_ListTextBlks.pdf](https://support.sas.com/content/dam/SAS/support/en/products-solutions/base-sas/tip-sheets/Tipsheet_ListTextBlks.pdf)**

**For ODS Tip Sheet:**

**<https://support.sas.com/content/dam/SAS/support/en/products-solutions/base-sas/tip-sheets/ods-tips.pdf>**

**For ODS LAYOUT Tip Sheet:**

**[https://support.sas.com/content/dam/SAS/support/en/products-solutions/base-sas/tip-sheets/Tipsheet\\_ods\\_layout.pdf](https://support.sas.com/content/dam/SAS/support/en/products-solutions/base-sas/tip-sheets/Tipsheet_ods_layout.pdf)**

**Full List of ODS tip sheets:**

**<https://support.sas.com/en/software/output-delivery-system-support/tip-sheets.html#6ddadff1-c328-4f03-b2fc-7a4393f76ca1>**

**Your questions, comments, and ideas about  
Output Delivery System (ODS) or about communicating with color, graphs,  
plots, charts, or are always welcome:**

LeRoy Bessler PhD

BesslerLeRoy2024@gmail.com

Bessler Consulting and Research

Visual Data Insights™

Strong Smart Systems™

Valparaiso, Indiana, USA

**Visual Data Insights is a trademark of LeRoy Bessler PhD.**